IN-32
1671
P. 39

NASA Technical Memorandum 103272

Coplanar Waveguide EEsoF MICAD Macros Make Circuit Layout Easy

George E. Ponchak
Lewis Research Center
Cleveland, Ohio

and

Nikola Visic
Cleveland State University
Cleveland, Ohio

September 1990

NASA

INDEX FOR

MICAD MACRO FILES

COPLANAR WAVEGUIDE EEsoF MICAD MACROS MAKE CIRCUIT LAYOUT EASY

George E. Ponchak
National Aeronautics and Space Administration
Lewis Research Center
Cleveland, Ohio 44135

and

Nikola Visic*
Cleveland State University
Cleveland, Ohio 44115

ABSTRACT

A collection of macro files is presented which permit the layout of coplanar waveguide integrated circuits using EEsoF MICAD CAD software. The files must be added to the users MICAD.ELE file.

INTRODUCTION

Coplanar waveguide (CPW) is a transmission line with a center strip conductor and a semi-infinite ground plane on either side [1]. Grounded coplanar waveguide (GCPW) has an additional ground plane on the back side of the substrate. Both of these transmission lines are highly suitable for low cost integrated circuit fabrication since the center conductor and the ground plane are on the same side of the substrate. This permits easy shunt as well as series connection of circuit elements without the need for wafer thinning and via hole fabrication. The difficulty with using CPW for integrated circuits has been the lack of circuit models and CAD layout tools.

EEsoF MICAD software is a CAD system for the layout and mask generation of etched microwave circuits. If the microwave circuit is to be fabricated by liftoff techniques or selective metal buildup as done in electroplating, then the negative of the generated mask must be used. Within MICAD, there is the ability to create templates to describe the CPW circuit elements. This is accomplished by creating macro files and appending them to the MICAD.ELE file [2]. The macro files are codes which describe the circuit element on a cartesian coordinate system. A series of macro files have been created to layout CPW elements. A diagram of the element, an example of how to incorporate the element into a EEsoF circuit file, and the macro file are presented in the format of the EEsoF element catalog in the EEsoF manuals. This should permit the addition of these macros into the users manual for easy use.

USE OF MICAD MACROS

The coplanar waveguide macros have been designed to work on plotting tables so that rubylith masks can be made. Not all of

---

*Co-op Student at NASA Lewis Research Center.

E-5714

the macros will work with other mask making systems. This is because coplanar waveguide has semi-infinite ground planes which are not closed surfaces. Therefore, the ground planes are defined as open surfaces. The circuit boundaries must be drawn by the user after the drawing is generated by Micad. In addition, Micad demands that the circuit nodes be defined on closed polygon surfaces. Although this is not a problem with other transmission line structures, this causes a problem with some CPW circuits which terminate the center conductor onto the ground planes which are defined as open surfaces. For these structures, the short circuit and short circuit terminated stub for example, an unwanted line will be plotted or cut into the rubylith. If care is taken not to peel the rubylith at this cut mark and a mask reduction is performed, the unwanted cut will not adversely effect the final mask.

The Micad macros may be typed into the MICAD.ELE file using the MICAD text editor or a word processor with an ASCII save feature. Although not stated in the MICAD manuals, there appears to be a size limit to the MICAD.ELE file length which can be read into the MICAD program. Therefore, it may be necessary to comment out the macro files in the MICAD.ELE file not required for the circuit being drawn. After the macros have been appended to the MICAD.ELE file, they may be used the same as "elements" are to layout a mask. An example of this is shown in Figures 1 and 2. Figure 1 is the EEsoF circuit file for a typical CPW 1 to 4 power divider with DC blocking capacitors and connection points for MMIC's. Figure 2 is the drawing for this circuit.

CONCLUSIONS

MICAD macro programs have been written which permit the layout of CPW circuits. These macros may be used with the elements supplied by EEsoF to create complex masks on a plotting machine.

REFERENCES

1. Wen, C.P.,"Coplanar Waveguide: A Surface Strip Transmission Line Suitable for Nonreciprocal Gyromagnetic Applications," IEEE Trans. Microwave Theory Tech., vol. MTT-17, no. 12, Dec. 1969, pp. 1087-1090.

2. EEsoF Micad Reference Manual.

```
dim
  freq ghz
  res oh
  ind nh
  cap pf
  lng mil
  time ps
  cond /oh
  ang deg
ckt
  cpw 1 2 w=12 g=10 l=214.5
  cstubo 2 3 w=12 g=12 os1=10 os2=12 os3=12 os4=20
  cpw 3 4 w=12 g=10 l=150
  ctaper 4 5 w1=12 g1=10 w2=40 g2=10 lt=103.5
  cpw 5 6 w=40 g=10 l=50
  ccplr 7 6 w=40 g=10 sc1=5 sc2=5 sc3=5 sc4=10 sc5=82.6
  cpw 7 8 w=40 g=10 l=50
  ctaper 8 9 w1=40 g1=10 w2=10 g2=10 lt=150
  ccp 9 10 w1=10 g1=10 l1=5 d=100 w2=10 g2=10 l2=5
  ctaper 10 11 w1=10 g1=10 w2=40 g2=10 lt=150
  cpw 11 12 w=40 g=10 l=50
  ccplr 12 13 w=40 g=10 sc1=5 sc2=5 sc3=5 sc4=10 sc5=82.6
  cpw 13 14 w=40 g=10 l=50
  ctaper 14 15 w1=40 g1=10 w2=12 g2=10 lt=100
  cpw 15 16 w=12 g=10 l=100
def2p 1 16 arm
  ctaper 1 2 w1=12 g1=10 w2=60 g2=10 lt=103.5
  ctee1 2 3 4 w1=60 g1=10 w2=12 g2=10 w3=12 g3=10
  cpw 3 5 w=12 g=10 l=141.5
  ccorn 5 6 w=12 g=10
  cpw 4 7 w=12 g=10 l=141.5
  ccorn 8 7 w=12 g=10
def3p 1 6 8 branch
  ctaper 1 2 w1=12 g1=10 w2=60 g2=10 lt=240
  ctee1 2 3 9 w1=60 g1=10 w2=12 g2=10 w3=12 g3=10
  cpw 3 4 w=12 g=10 l=150
  cstubg 4 5 w=12 g=10 zg1=29 zg2=10
  cpw 5 6 w=12 g=10 l=150
  ccorn 6 7 w=12 g=10
  cpw 7 8 w=12 g=10 l=114.5
  cpw 9 11 w=12 g=10 l=150
  cstubg 11 12 w=12 g=10 zg1=29 zg2=10
  cpw 12 13 w=12 g=10 l=150
  ccorn 14 13 w=12 g=10
  cpw 14 15 w=12 g=10 l=114.5
  branch 8 50 51
  branch 15 52 53
  arm 50 54
  arm 51 55
  arm 52 56
  arm 53 57
def5p 1 54 55 56 57 print
```

Figure 1. MICAD circuit file for a CPW 1 to 4 power divider.



Figure 2. CPW 1 to 4 Power divider.

CBEND Coplanar Waveguide Bend CBEND

Physical layout:



Data:

W=Strip width
G=Slot width
ANG=Angle of bend

Syntax:

```
CBEND  n1 n2 W=x1 G=x2 ANG=x3
```

Example:

```
CBEND 1 2 W=25 G=10 ANG=40
```

Notes:

1. This is a macro program in the MICAD.ELE file.
2. The program is valid for 0<=ANG<=180 degree.

```
defelem "CBEND",2,"W","G","ANG"
         dim ANGRAD,TN,CS,SN
         ANGRAD=ANGUNIT*ANG
         TN=tan(ANGRAD/2.0)
         CS=cos(ANGRAD)
         SN=sin(ANGRAD)
         level lmet1
         point 4,0,W/2.0
         node n1,0,0
         point 8,0,-W/2.0
         point 8,G*TN,-W/2.0
         point 8,G*SN,-(W/2.0+G)+G*CS
         node n2,(W/2.0+G)*SN,-(W/2.0+G)+(W/2.0+G)*CS
         point 8,(W+G)*SN,-(W/2.0+G)+(W+G)*CS
         point 8,(W+G)*TN,W/2.0
         point 12,0,W/2.0
         point 3,0,-(W/2.0+G)
         point 11,0,-(W/2.0+G)
         point 3,0,W/2.0+G
         point 8,(W+2.0*G)*TN,W/2.0+G
         point 11,(W+2.0*G)*SN,-(W/2.0+G)+(W+2.0*G)*CS
end define
```

Program written by George E. Ponchak

CCORN Coplanar Waveguide Corner CCORN

Physical layout:



Data:
W=Strip width
G=Slot width

Syntax:
CCORN n1 n2 W=x1 G=x2

Example:
CCORN 1 2 W=25 G=10

Notes:
This is a macro program in the MICAD.ELE file.

```
defelem "CCORN",2,"W","G"
         level lmet1
         point 4,0,w/2
         node n1,0,0
         point 8,0,-w/2
         point 8,g,-w/2
         point 8,g,-w/2-g
         node n2,g+w/2,-w/2-g
         point 8,g+w,-w/2-g
         point 8,g+w,w/2
         point 12,0,w/2
         point 3,w+2*g,-w/2-g
         point 8,w+2*g,w/2+g
         point 11,0,w/2+g
         point 3,0,-w/2-g
         point 11,0,-w/2-g
end define
```

Program written by Nikola Visic

CCP Coplanar Waveguide Connection Point CCP

Physical layout:



Data:

W1=Strip width at node 1
G1=Slot width at node 1
W2=Strip width at node 2
G2=Slot width at node 2
L1=Open end length at node 1
L2=Open end length at node 2
D=Connection point length

Syntax:

CCP n1 n2 W1=x1 G1=x2 L1=x3 D=x4 W2=x5 G2=x6 L2=x7

Example:

CCP 1 2 W1=25 G1=10 L1=10 D=50 W2=20 G2=8 L2=10

Notes:

1. This is a macro program in the MICAD.ELE file.

2. This element is useful for integrating circuit elements which require a ground connection such as transistors or microstrip based integrated circuits.

```
defelem "CCP",2,"W1","G1","L1","D","W2","G2","L2"
        level lmet1
        point 4,0,W1/2
        node n1,0,0
        point 12,0,-W1/2
        point 3,0,W1/2+G1
        point 8,L1,W1/2+G1
        point 8,L1,-W1/2-G1
        point 11,0,-W1/2-G1
        point 4,L1+D+L2,-W2/2
        node n2,L1+D+L2,0
        point 12,L1+D+L2,W2/2
        point 3,L1+D+L2,W2/2+G2
        point 8,L1+D,W2/2+G2
        point 8,L1+D,-W2/2-G2
        point 11,L1+D+L2,-W2/2-G2
end define
```

Program written by Nikola Visic

CCPLR Coplanar Waveguide Interdigital Coupler CCPLR

Physical layout:



Data:

W=Strip width
G=Slot width
SC1=Slot width
SC2=Slot width
SC3=Slot width
SC4=Center conductor width in the coupler region
SC5=Length of the center conductor in the coupler region

Syntax:

CCPLR n1 n2 W=x1 G=x2 SC1=x3 SC2=x4 SC3=x5 SC4=x6 SC5=x7

Example:

CCPLR 1 2 W=25 G=10 SC1=10 SC2=5 SC3=5 SC4=8 SC5=50

Notes:

1. This is a macro program in the MICAD.ELE file.

2. The program does not run a check on the validity of the data. It is up to the user to check that 2*SC2+SC4<W.

```
defelem "CCPLR",2,"W","G","Sc1","Sc2","Sc3","Sc4","Sc5"
         level lmet1
         point 4,0,W/2.0
         node n1,0,0
         point 8,0,-W/2.0
         point 8,0,-Sc4/2.0
         point 8,Sc5,-Sc4/2.0
         point 8,Sc5,Sc4/2.0
         point 8,0,Sc4/2.0
         point 8,0,W/2.0
         point 12,0,W/2.0
         point 3,0,-(W/2.0+G)
         point 11,(Sc5+Sc3),-(W/2.0+G)
         point 4,(Sc5+Sc3),-W/2.0
         node n2,(Sc5+Sc3),0
         point 8,(Sc5+Sc3),W/2.0
         point 8,Sc1,W/2.0
         point 8,Sc1,(Sc4/2.0+Sc2)
         point 8,(Sc5+Sc3),(Sc4/2.0+Sc2)
         point 8,(Sc5+Sc3),-(Sc4/2.0+Sc2)
         point 8,Sc1,-(Sc4/2.0+Sc2)
         point 8,Sc1,-W/2.0
         point 12,(Sc5+Sc3),-W/2.0
         point 3,(Sc5+Sc3),(W/2.0+G)
         point 11,0,(W/2.0+G)
end define
```

Program written by George E. Ponchak

Physical layout:



Data:

W=Strip width
G=Slot width
ANG=Angle through which the curve turns
RAD=Radius of curve

Syntax:

CCURVE n1 n2 W=x1 G=x2 ANG=x3 RAD=x4

Example:

CCURVE 1 2 W=25 G=10 ANG=70 RAD=30

Notes:

1.This is a macro program in the MICAD.ELE file.

2.The program does not check the validity of the data. It is up to the user to check that 0<ANG<180 degrees and RAD>W/2.0+G.

```
defelem "CCURVE",2,"W","G","ANG","RAD"
         dim PI,RADTODEG,ANGRAD,ANGDEG,RADGIN,RADWIN,RADWOUT
         dim B,RADGOUT
         level lmet1
         PI=3.141592654
         RADTODEG=360./(2.0*PI)
         ANGRAD=-ANGUNIT*ANG
         ANGDEG=ANGRAD*RADTODEG
         RADGIN=RAD-(W/2.0+G)
         RADWIN=RAD-W/2.0
         RADWOUT=RAD+W/2.0
         RADGOUT=RAD+W/2.0+G
         B=-PI+ANGRAD
         point 4,0,W/2.0
         node n1,0,0
         point 8,0,-W/2.0
         point 9,RADWIN,ANGDEG
         point 10,0,-RAD
         point 8,RADWIN*sin(B),-RAD-RADWIN*cos(B)
         node n2,RAD*sin(B),-RAD*(1.0+cos(B))
         point 8,RADWOUT*sin(B),-RAD-RADWOUT*cos(B)
         point 9,RADWOUT,-ANGDEG
         point 10,0,-RAD
         point 12,0,W/2.0
         point 3,0,-(W/2.0+G)
         point 9,RADGIN,ANGDEG
         point 10,0,-RAD
         point 11,RADGIN*sin(B),-RAD-RADGIN*cos(B)
         point 3,0,W/2.0+G
         point 9,RADGOUT,ANGDEG
         point 10,0,-RAD
         point 11,RADGOUT*sin(B),-RAD-RADGOUT*cos(B)
end define
```

Program written by George E. Ponchak

CGAP Coplanar Waveguide Gap CGAP

Physical layout:



Data:

W=Strip width
G=Slot width
SG=Gap width

Syntax:

```
CGAP n1 n2 W=x1 G=x2 SG=x3
```

Example:

```
CGAP 1 2 W=25 G=10 SG=15
```

Notes:

This is a macro program in the MICAD.ELE file.

```
defelem "CGAP",2,"W","G","Sg"
        level lmet1
        point 4,0,W/2.0
        node n1,0,0
        point 12,0,-W/2.0
        point 3,0,-(W/2.0+G)
        point 11,Sg,-(W/2.0+G)
        point 4,Sg,-W/2.0
        node n2,Sg,0
        point 12,Sg,W/2.0
        point 3,Sg,(W/2.0+G)
        point 11,0,(W/2.0+G)
end define
```

Program written by George E. Ponchak

Physical layout:



Data:
W=Strip width
G=Slot width
SO=Open end width

Syntax:
COPEN n1 W=x1 G=x2 SO=x3

Example:
COPEN 1 W=25 G=10 SO=10

Notes:
This is a macro program in the MICAD.ELE file.

```
defelem "COPEN",1,"W","G","So"
        level lmet1
        point 4,0,W/2.0
        node n1,0,0
        point 12,0,-W/2.0
        point 3,0,-(W/2.0+G)
        point 8,So,-(W/2.0+G)
        point 8,So,W/2.0+G
        point 11,0,W/2.0+G
end define
```

Program written by George E. Ponchak

CSHORT Coplanar Waveguide Short Circuit CSHORT

Physical layout:



Data:
W=Strip width
G=Slot width

Syntax:
CSHORT n1 W=x1 G=x2

Example:
CSHORT 3 W=25 G=10

Notes:
1. This is a macro program in the MICAD.ELE file.

2. This macro will draw an unwanted line at the node. If care is taken not to lift the rubylith mask at this line and a mask reduction is performed, then the line should not effect the final mask.

```
defelem "CSHORT",1,"W","G"
          level lmet1
          point 3,0,W/2.0+G
          point 11,0,W/2.0
          point 4,0,W/2.0
          node n1,0,0
          point 12,0,-W/2.0
          point 3,0,-W/2.0
          point 11,0,-(W/2.0+G)
end define
```

Program written by George E. Ponchak

CSPURC Coplanar Waveguide Spurline Filter CSPURC
in the Center Conductor

Physical layout:



Data:

W=Strip width
G=Slot width
SS1=Slot width
SS2=Slot width
SS3=Length of the filter
SS4=Center conductor width in the filter region

Syntax:

CSPURC n1 n2 W=x1 G=x2 SS1=x3 SS2=x4 SS3=x5 SS4=x6

Example:

CSPURC 1 2 W=25 G=10 SS1=10 SS2=5 SS3=50 SS4=8

Notes:

1. This is a macro program in the MICAD.ELE file.

2. The program does not run a check on the validity of the data. It is up to the user to check that 2*SS2+SS4<W.

```
defelem "CSPURC",2,"W","G","Ss1","Ss2","Ss3","Ss4"
        level lmet1
        point 4,0,W/2.0
        node n1,0,0
        point 8,0,-W/2.0
        point 8,0,-Ss4/2.0
        point 8,Ss3,-Ss4/2.0
        point 8,Ss3,-(Ss4/2.0+Ss2)
        point 8,Ss1,-(Ss4/2.0+Ss2)
        point 8,Ss1,-W/2.0
        point 8,Ss3,-W/2.0
        node n2,Ss3,0
        point 8,Ss3,W/2.0
        point 8,Ss1,W/2.0
        point 8,Ss1,(Ss4/2.0+Ss2)
        point 8,Ss3,(Ss4/2.0+Ss2)
        point 8,Ss3,Ss4/2.0
        point 8,0,Ss4/2.0
        point 8,0,W/2.0
        point 12,0,W/2.0
        point 3,0,-(W/2.0+G)
        point 11,Ss3,-(W/2.0+G)
        point 3,Ss3,(W/2.0+G)
        point 11,0,(W/2.0+G)
end define
```

Program written by George E. Ponchak

CSPURG Coplanar Waveguide Spurline Filter in the Ground Plane CSPURG

Physical layout:



Data:

W=Strip width
G=Slot width
SSG1=Slot width
SSG2=Distance between filter slot and CPW slot
SSG3=Slot width
SSG4=Length of the filter region

Syntax:

CSPURG n1 n2 W=x1 G=x2 SSG1=x3 SSG2=x4 SSG3=x5 SSG4=x6

Example:

CSPURG 1 2 W=25 G=10 SSG1=10 SSG2=5 SSG3=8 SSG4=50

Notes:

This is a macro program in the MICAD.ELE file.

```
defelem "CSPURG",2,"W","G","Ssg1","Ssg2","Ssg3","Ssg4"
        level lmet1
        point 4,0,W/2.0
        node n1,0,0
        point 8,0,-W/2.0
        point 8,Ssg4,-W/2.0
        node n2,Ssg4,0
        point 8,Ssg4,W/2.0
        point 12,0,W/2.0
        point 3,0,-(W/2.0+G)
        point 8,0,-(W/2.0+G+Ssg2+Ssg3)
        point 8,Ssg4,-(W/2.0+G+Ssg2+Ssg3)
        point 8,Ssg4,-(W/2.0+G+Ssg2)
        point 8,Ssg1,-(W/2.0+G+Ssg2)
        point 8,Ssg1,-(W/2.0+G)
        point 11,Ssg4,-(W/2.0+G)
        point 3,Ssg4,(W/2.0+G)
        point 8,Ssg1,(W/2.0+G)
        point 8,Ssg1,(W/2.0+G+Ssg2)
        point 8,Ssg4,(W/2.0+G+Ssg2)
        point 8,Ssg4,(W/2.0+G+Ssg2+Ssg3)
        point 8,0,(W/2.0+G+Ssg2+Ssg3)
        point 11,0,(W/2.0+G)
end define
```

Program written by George E. Ponchak

CSTART Coplanar Waveguide Open at Input CSTART

Physical layout:



Data:

W=Strip width
G=Slot width
SOS=Open end at input

Syntax:

```
CSTART n1 n2 W=x1 G=x2 SOS=x3
```

Example:

```
CSTART 1 2 W=25 G=10 SOS=55
```

Notes:

1. This is a macro program in the MICAD.ELE file.

2. This structure is useful for wafer probing CPW cicuits and for starting CPW circuits in the interior of a substrate.

```
defelem "CSTART",2,"W","G","Sos"
        level lmet1
        point 3,Sos,(W/2.0+G)
        point 11,0,(W/2.0+G)
        point 4,0,(W/2.0+G)
        node n1,0,0
        point 12,0,-(W/2.0+G)
        point 3,0,-(W/2.0+G)
        point 11,Sos,-(W/2.0+G)
        point 4,Sos,-W/2.0
        node n2,Sos,0
        point 12,Sos,W/2.0
end define
```

Program written by George E. Ponchak

CSTARTP Coplanar Waveguide Open at Input with Pads for RF Wafer Probes CSTARTP

Physical layout:



Data:

W=Strip width
G=Slot width
SOS=Open end at input
P=Pitch of wafer probes
Pad=Side of pad

Syntax:

CSTARTP n1 n2 W=x1 G=x2 SOS=x3 P=x4 Pad=x5

Example:

CSTART 1 2 W=25 G=10 SOS=55 P=10 Pad=4

Notes:

1. This is a macro program in the MICAD.ELE file.

2. This structure is useful for wafer probing CPW cicuits and for starting CPW circuits in the interior of a substrate.

3. Cascade Microtech recommends 4 by 4 mil pads (100 by 100 micron). The minimum recommended pad is 2 by 2 mil.

4. The pads are defined on the dielectric layer mask level. This may be changed in the program at the commented line.

```
defelem "CSTARTP",2,"W","G","Sos","P","Pad"
        level lmet1
        point 3,Sos+Pad,(W/2.0+G)
        point 11,0,(W/2.0+G)
        point 4,0,(W/2.0+G)
        node n1,0,0
        point 12,0,-(W/2.0+G)
        point 3,0,-(W/2.0+G)
        point 11,Sos+Pad,-(W/2.0+G)
        point 4,Sos+Pad,-W/2.0
        node n2,Sos+Pad,0
        point 8,Sos+Pad,W/2.0
        point 8,Sos,W/2.0
        point 8,Sos,-W/2.0
        point 12,Sos+Pad,-W/2.0
        level ldiel ! defines mask layer for pad definition
        point 5,Sos,Pad/2.0
        point 8,Sos,-Pad/2.0
        point 8,Sos+Pad,-Pad/2.0
        point 8,Sos+Pad,Pad/2.0
        point 12,Sos,Pad/2.0
        point 5,Sos,P+Pad/2.0
        point 8,Sos,P-Pad/2.0
        point 8,Sos+Pad,P-Pad/2.0
        point 8,Sos+Pad,P+Pad/2.0
        point 12,Sos,P+Pad/2.0
        point 5,Sos,-P+Pad/2.0
        point 8,Sos,-P-Pad/2.0
        point 8,Sos+Pad,-P-Pad/2.0
        point 8,Sos+Pad,-P+Pad/2.0
        point 12,Sos,-P+Pad/2.0
end define

Program written by George E. Ponchak
```

CSTEP Coplanar Waveguide Step CSTEP

Physical layout:



Data:

W1=Strip width at node 1
G1=Slot width at node 1
W2=Strip width at node 2
G2=Slot width at node 2

Syntax:

CSTEP n1 n2 W1=x1 G1=x2 W2=x3 G2=x4

Example:

CSTEP 1 2 W1=25 G1=10 W2=15 G2=20

Notes:

1. This is a macro program in the MICAD.ELE file.

2. There are no limitations on W1, W2, G1, G2. Any of the data inputs may be varied to create a step in the center conductor, the ground planes, or both. (W1<W2+2*G2, W2<W1+2*G1)

```
defelem "CSTEP",2,"W1","G1","W2","G2"
        level lmet1
        point 4,0,W1/2.0
        node n1,0,0
        point 8,0,-W1/2.0
        point 8,0,-W2/2.0
        node n2,0,0
        point 8,0,W2/2.0
        point 12,0,W1/2.0
        point 3,0,-(W1/2.0+G1)
        point 11,0,-(W2/2.0+G2)
        point 3,0,(W1/2.0+G1)
        point 11,0,(W2/2.0+G2)
end define
```

Program written by George E. Ponchak

Physical layout:



Data:

W=Strip width
G=Slot width
ZC1=Stub slot width
ZC2=Stub slot length

Syntax:

CSTUBC n1 n2 W=x1 G=x2 ZC1=x3 ZC2=x4

Example:

CSTUBG 1 2 W=25 G=10 ZC1=5 ZC2=10

Notes:

1. This is a macro program in the MICAD.ELE file.

2. The program does not run a check on the validity of the data. It is up to the user to check that 2*ZC2<W.

```
defelem "CSTUBC",2,"W","G","Zc1","Zc2"
        level lmet1
        point 4,0,W/2.0
        node n1,0,0
        point 8,0,-W/2.0
        point 8,0,(-W/2.0+Zc2)
        point 8,Zc1,(-W/2.0+Zc2)
        point 8,Zc1,-W/2.0
        node n2,Zc1,0
        point 8,Zc1,W/2.0
        point 8,Zc1,(W/2.0-Zc2)
        point 8,0,(W/2.0-Zc2)
        point 12,0,W/2.0
        point 3,0,-(W/2.0+G)
        point 11,Zc1,-(W/2.0+G)
        point 3,Zc1,(W/2.0+G)
        point 11,0,(W/2.0+G)
end define
```

Program written by George E. Ponchak

CSTUBG Coplanar Waveguide Slot Stub in the Ground Plane CSTUBG

Physical layout:



Data:

W=Strip width
G=Slot width
ZG1=Stub slot width
ZG2=Stub slot length

Syntax:

```
CSTUBG n1 n2 W=x1 G=x2 ZG1=x3 ZG2=x4
```

Example:

```
CSTUBG 1 2 W=25 G=10 ZG1=5 ZG2=40
```

Notes:

This is a macro program in the MICAD.ELE file.

```
defelem "CSTUBG",2,"W","G","Zg1","Zg2"
        level lmet1
        point 4,0,W/2.0
        node n1,0,0
        point 8,0,-W/2.0
        point 8,Zg1,-W/2.0
        node n2,Zg1,0
        point 8,Zg1,W/2.0
        point 12,0,W/2.0
        point 3,0,-(W/2.0+G)
        point 8,0,-(W/2.0+G+Zg2)
        point 8,Zg1,-(W/2.0+G+Zg2)
        point 11,Zg1,-(W/2.0+G)
        point 3,Zg1,(W/2.0+G)
        point 8,Zg1,(W/2.0+G+Zg2)
        point 8,0,(W/2.0+G+Zg2)
        point 11,0,(W/2.0+G)
end define
```

Program written by George E. Ponchak

CSTUBO Coplanar Waveguide Open Ended Stub CSTUBO

Physical layout:



Data:

W=Strip width
G=Slot width
OS1=Stub width
OS2=Slot width of stub
OS3=Open end width
OS4=Stub length

Syntax:

CSTUBO n1 n2 W=x1 G=x2 OS1=x3 OS2=x4 OS3=x5 OS4=x6

Example:

CSTUBO 1 2 W=25 G=10 OS1=20 OS2=10 OS3=10 OS4=30

Notes:

This is a macro program in the MICAD.ELE file.

```
defelem "CSTUBO",2,"W","G","OS1","OS2","OS3","OS4"
          dim OS5
          dim OS6
          OS5=OS1+2.0*OS2
          OS6=OS1+OS2
          level lmet1
          point 4,0,W/2.0
          node n1,0,0
          point 8,0,-W/2.0
          point 8,OS2,-W/2.0
          point 8,OS2,-(W/2.0+OS4)
          point 8,OS6,-(W/2.0+OS4)
          point 8,OS6,-W/2.0
          point 8,OS5,-W/2.0
          node n2,OS5,0
          point 8,OS5,W/2.0
          point 8,OS6,W/2.0
          point 8,OS6,(W/2.0+OS4)
          point 8,OS2,(W/2.0+OS4)
          point 8,OS2,W/2.0
          point 12,0,W/2.0
          point 3,0,-(W/2.0+G)
          point 8,0,-(W/2.0+OS3+OS4)
          point 8,OS5,-(W/2.0+OS3+OS4)
          point 11,OS5,-(W/2.0+G)
          point 3,OS5,(W/2.0+G)
          point 8,OS5,(W/2.0+OS3+OS4)
          point 8,0,(W/2.0+OS3+OS4)
          point 11,0,(W/2.0+G)
end define
```

Program written by George E. Ponchak

CSTUBS Coplanar Waveguide Short Circuit Terminated Stub CSTUBS

Physical layout:



Data:

W=Strip width
G=Slot width
WS=Strip width of stub
GS=Slot width of stub
S=Stub length

Syntax:

CSTUBS n1 n2 W=x1 G=x2 GS=x3 WS=x4 S=x5

Example:

CSTUBS 1 2 W=25 G=10 GS=8 WS=16 S=30

Notes:

1. This is a macro program in the MICAD.ELE file.

2. This macro will draw two unwanted lines at the nodes. If care is taken not to lift the rubylith mask at these lines and a mask reduction is performed, then these lines should not effect the final mask.

```
defelem "CSTUBS",2,"W","G","GS","WS","S"
          dim X
          level lmet1
          X=2.0*GS+WS
          point 4,0,W/2.0
          node n1,0,0
          point 12,0,-W/2.0
          point 4,X,-W/2.0
          node n2,X,0
          point 12,X,W/2.0
          point 3,0,-(W/2.0+G)
          point 8,0,-(W/2.0+S)
          point 8,GS,-(W/2.0+S)
          point 8,GS,-W/2.0
          point 11,0,-W/2.0
          point 3,X,-W/2.0
          point 8,GS+WS,-W/2.0
          point 8,GS+WS,-(W/2.0+S)
          point 8,X,-(W/2.0+S)
          point 11,X,-(W/2.0+G)
          point 3,X,W/2.0+G
          point 8,X,W/2.0+S
          point 8,GS+WS,W/2.0+S
          point 8,GS+WS,W/2.0
          point 11,X,W/2.0
          point 3,0,W/2.0
          point 8,GS,W/2.0
          point 8,GS,W/2.0+S
          point 8,0,W/2.0+S
          point 11,0,W/2.0+G
end define
```

Program written by George E. Ponchak

CTAPER Coplanar Waveguide Taper CTAPER

Physical layout:



Data:

W1=Strip width at node 1
G1=Slot width at node 1
W2=Strip width at node 2
G2=slot width at node 2
LT=Taper length

Syntax:

```
CTAPER n1 n2 W1=x1 G1=x2 W2=x3 G2=x4 LT=x5
```

Example:

```
CTAPER 1 2 W1=25 G1=10 W2=15 G2=5 LT=100
```

Notes:

1. This is a macro program in the MICAD.ELE file.

2. There are no limitations on W1, W2, G1, G2. Any of the data inputs may be varied to creat a taper of the center conductor, the slots, or both.

```
defelem "CTAPER",2,"W1","G1","W2","G2","LT"
        level lmet1
        point 4,0,W1/2.0
        node n1,0,0
        point 8,0,-W1/2.0
        point 8,LT,-W2/2.0
        node n2,LT,0
        point 8,LT,W2/2.0
        point 12,0,W1/2.0
        point 3,0,-(W1/2.0+G1)
        point 11,Lt,-(W2/2.0+G2)
        point 3,Lt,(W2/2.0+G2)
        point 11,0,(W1/2.0+G1)
end define
```

Program written by George E. Ponchak

Physical layout:



Data:

W1=Strip width at node 1
G1=Slot width at node 1
W2=Strip width at node 2
G2=Slot width at node 2
W3=Strip width at node 3
G3=Slot width at node 3

Syntax:

CTEE n1 n2 n3 W1=x1 G1=x2 W2=x3 G2=x4 W3=x5 G3=x6

Example:

CTEE 1 2 3 W1=25 G1=10 W2=20 G2=10 W3=30 G3=5

Notes:

1. This is a macro program in the MICAD.ELE file.

2. This Tee junction aligns the slot opposite node 1. Nodes 2 and 3 may not be aligned.

```
 defelem "CTEE1",3,"W1","G1","W2","G2","W3","G3"
        dim d
        if 2*G3+W3-2*G2-W2 > 0 then
          d=0
         else
          d=2*G3+W3-2*G2-W2
        end if
        level lmet1
        point 4,-W1/2,d
        node n1,0,d
        point 8,W1/2,d
        point 8,W1/2,G3
        point 8,W1/2+G1,G3
        node n3,W1/2+G1,G3+W3/2
        point 8,W1/2+G1,G3+W3
        point 8,0,G3+W3
        point 8,0,2*G3+W3-G2
        point 8,-(W1/2+G1),2*G3+W3-G2
        node n2,-(W1/2+G1),2*G3+W3-G2-W2/2
        point 8,-(W1/2+G1),2*G3+W3-G2-W2
        point 8,-W1/2,2*G3+W3-G2-W2
        point 12,-W1/2,d
        point 3,W1/2+G1,d
        point 11,W1/2+G1,0
        point 3,-(W1/2+G1),2*G3+W3-2*G2-W2
        point 11,-(W1/2+G1),d
        point 3,W1/2+G1,2*G3+W3
        point 11,-(W1/2+G1),2*G3+W3
end define

Program written by Nikola Visic
```

Physical layout:



Data:

W1=Strip width at node 1
G1=Slot width at node 1
W2=Strip width at node 2
G2=Slot width at node 2
W3=Strip width at node 3
G3=Slot width at node 3

Syntax:

```
CTEE n1 n2 n3 W1=x1 G1=x2 W2=x3 G2=x4 W3=x5 G3=x6
```

Example:

```
CTEE 1 2 3 W1=25 G1=10 W2=20 G2=10 W3=30 G3=5
```

Notes:

1. This is a macro program in the MICAD.ELE file.

2. This Tee junction aligns nodes 2 and 3.

```
defelem "CTEE2",3,"W1","G1","W2","G2","W3","G3"
        dim d2,d3,d
        d=G2+W2/2.0-(G3+W3/2.0)
        if d<=0 then
        d2=-d
        d3=0.0
        else
        d2=0.0
        d3=d
        end if
        level lmet1
        point 4,0,W1/2.0
        node n1,0,0
        point 8,0,-W1/2.0
        point 8,d3+G3,-W1/2.0
        point 8,d3+G3,-(W1/2.0+G1)
        node n3,d3+G3+W3/2.0,-(W1/2.0+G1)
        point 8,d3+G3+W3,-(W1/2.0+G1)
        point 8,d3+G3+W3,0
        point 8,d2+G2+W2,0
        point 8,d2+G2+W2,W1/2.0+G1
        node n2,d2+G2+W2/2.0,W1/2.0+G1
        point 8,d2+G2,W1/2.0+G1
        point 8,d2+G2,W1/2.0
        point 12,0,W1/2.0
        point 3,0,W1/2.0+G1
        point 11,d2,W1/2.0+G1
        point 3,0,-(W1/2.0+G1)
        point 11,d3,-(W1/2.0+G1)
        point 3,d3+2.0*G3+W3,-(W1/2.0+G1)
        point 8,d3+2.0*G3+W3,0
        point 8,d2+2.0*G2+W2,0
        point 11,d2+2.0*G2+W2,W1/2.0+G1
end define
```

Program written by George E. Ponchak

NASA
National Aeronautics and Space Administration

Report Documentation Page

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA TM-103272		

4. Title and Subtitle	5. Report Date
Coplanar Waveguide EEsoF MICAD Macros Make Circuit Layout Easy	September 1990
	6. Performing Organization Code

7. Author(s)	8. Performing Organization Report No.
George E. Ponchak and Nikola Visic	E-5714
	10. Work Unit No.
	506-44-2B

9. Performing Organization Name and Address	11. Contract or Grant No.
National Aeronautics and Space Administration Lewis Research Center Cleveland, Ohio 44135-3191	
	13. Type of Report and Period Covered
	Technical Memorandum

12. Sponsoring Agency Name and Address	14. Sponsoring Agency Code
National Aeronautics and Space Administration Washington, D.C. 20546-0001	

15. Supplementary Notes

George E. Ponchak, Lewis Research Center. Nikola Visic, Cleveland State University, Cleveland, Ohio 44115, and Co-op student at Lewis Research Center.

16. Abstract

A collection of macro files is presented which permit the layout of coplanar waveguide integrated circuits using EEsoF MICAD CAD software. The files must be added to the users MICAD.ELE file.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Microwave Transmission lines Integrated circuits	Unclassified – Unlimited Subject Category 32

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of pages	22. Price*
Unclassified	Unclassified	48	A03

NASA FORM 1626 OCT 86

*For sale by the National Technical Information Service, Springfield, Virginia 22161